QT += network xml
QT -= gui

LYK_INCLUDE_DIR = $$PWD/../../include
LYK_LIBRARY_DIR = $$PWD/../../lib
LYK_LIBRARY_VERSION = 1.0

CONFIG(debug, debug|release) {
    LYK_LIBRARY_NAME = lyk_lib_d
} else {
    LYK_LIBRARY_NAME = lyk_lib
}

TEMPLATE = lib

CONFIG += staticlib
INCLUDEPATH += $$LYK_LIBRARY_DIR


# Target definition
TARGET = $$LYK_LIBRARY_NAME
VERSION = $$LYK_LIBRARY_VERSION
DESTDIR = $$LYK_LIBRARY_DIR

# Header files
INSTALL_HEADERS = $$LYK_INCLUDE_DIR/FormPost.h \
    $$LYK_INCLUDE_DIR/ConfigSettings.h

HEADERS += $$INSTALL_HEADERS

# Source files
SOURCES += FormPost.cpp \
    ConfigSettings.cpp

# Installation
headers.files = $$INSTALL_HEADERS
headers.path = $$[QT_INSTALL_PREFIX]/include/lyk_qt
target.path = $$[QT_INSTALL_LIBS]
INSTALLS += headers target

# pkg-config support
CONFIG += create_pc create_prl no_install_prl
QMAKE_PKGCONFIG_DESTDIR = $$LYK_LIBRARY_DIR/pkgconfig
QMAKE_PKGCONFIG_LIBDIR = $$target.path
QMAKE_PKGCONFIG_INCDIR = $$headers.path
